**Description**

`ttest` performs \( t \) tests on the equality of means. The test can be performed for one sample against a hypothesized population mean. Two-sample tests can be conducted for paired and unpaired data. The assumption of equal variances can be optionally relaxed in the unpaired two-sample case.

`ttesti` is the immediate form of `ttest`; see [U] 19 Immediate commands.

**Quick start**

Test that the mean of `v1` is equal between two groups defined by `catvar`

```
ttest v1, by(catvar)
```

Same as above, but assume unequal variances
```
ttest v1, by(catvar) unequal
```

Paired \( t \) test of `v2` and `v3`
```
ttest v2 == v3
```

Same as above, but with unpaired data and conduct test separately for each level of `catvar`
```
by catvar: ttest v2 == v3, unpaired
```

Test that the mean of `v4` is 3 at the 90% confidence level
```
ttest v4 == 3, level(90)
```

Test \( \mu_1 = \mu_2 \) if \( \bar{x}_1 = 3.2, \sigma_1 = 0.1, \bar{x}_2 = 3.4, \) and \( \sigma_2 = 0.15 \) with \( n_1 = n_2 = 12 \)
```
ttesti 12 3.2 .1 12 3.4 .15
```

**Menu**

`ttest`

Statistics > Summaries, tables, and tests > Classical tests of hypotheses > t test (mean-comparison test)

`ttesti`

Statistics > Summaries, tables, and tests > Classical tests of hypotheses > t test calculator
**Syntax**

One-sample t test

\[ \texttt{ttest varname == # [if] [in] [, level(#)]} \]

Two-sample t test using groups

\[ \texttt{ttest varname [if] [in], by(groupvar) [options_1]} \]

Two-sample t test using variables

\[ \texttt{ttest varname_1 == varname_2 [if] [in], unpaired [unequal welch level(#)]} \]

Paired t test

\[ \texttt{ttest varname_1 == varname_2 [if] [in] [, level(#)]} \]

Immediate form of one-sample t test

\[ \texttt{ttesti #obs #mean #sd #val [, level(#)]} \]

Immediate form of two-sample t test

\[ \texttt{ttesti #obs1 #mean1 #sd1 #obs2 #mean2 #sd2 [, options_2]} \]

<table>
<thead>
<tr>
<th>options_1</th>
<th>Description</th>
</tr>
</thead>
<tbody>
<tr>
<td>by(groupvar)</td>
<td>variable defining the groups</td>
</tr>
<tr>
<td>reverse</td>
<td>reverse group order for mean difference computation</td>
</tr>
<tr>
<td>unequal</td>
<td>unpaired data have unequal variances</td>
</tr>
<tr>
<td>welch</td>
<td>use Welch’s approximation</td>
</tr>
<tr>
<td>level(#)</td>
<td>set confidence level; default is level(95)</td>
</tr>
</tbody>
</table>

*by(groupvar) is required.

<table>
<thead>
<tr>
<th>options_2</th>
<th>Description</th>
</tr>
</thead>
<tbody>
<tr>
<td>unequal</td>
<td>unpaired data have unequal variances</td>
</tr>
<tr>
<td>welch</td>
<td>use Welch’s approximation</td>
</tr>
<tr>
<td>level(#)</td>
<td>set confidence level; default is level(95)</td>
</tr>
</tbody>
</table>

by and collect are allowed with ttest and ttesti; see [U] 11.1.10 Prefix commands.
Options

by(groupvar) specifies the groupvar that defines the two groups that ttest will use to test the hypothesis that their means are equal. Specifying by(groupvar) implies an unpaired (two sample) t test. Do not confuse the by() option with the by prefix; you can specify both.

reverse reverses the order of the mean difference between groups defined in by(). By default, the mean of the group corresponding to the largest value in the variable in by() is subtracted from the mean of the group with the smallest value in by(). reverse reverses this behavior and the order in which variables appear on the table.

unpaired specifies that the data be treated as unpaired. The unpaired option is used when the two sets of values to be compared are in different variables.

unequal specifies that the unpaired data not be assumed to have equal variances.

welch specifies that the approximate degrees of freedom for the test be obtained from Welch’s formula (1947) rather than from Satterthwaite’s approximation formula (1946), which is the default when unequal is specified. Specifying welch implies unequal.

level(#) specifies the confidence level, as a percentage, for confidence intervals. The default is level(95) or as set by set level; see [U] 20.8 Specifying the width of confidence intervals.

Remarks and examples

Remarks are presented under the following headings:

- One-sample t test
- Two-sample t test
- Paired t test
- Two-sample t test compared with one-way ANOVA
- Immediate form
- Video examples

One-sample t test

Example 1

In the first form, ttest tests whether the mean of the sample is equal to a known constant under the assumption of unknown variance. Assume that we have a sample of 74 automobiles. We know each automobile’s average mileage rating and wish to test whether the overall average for the sample is 20 miles per gallon.
. use https://www.stata-press.com/data/r18/auto
(1978 automobile data)
. ttest mpg==20

One-sample t test

<table>
<thead>
<tr>
<th>Variable</th>
<th>Obs</th>
<th>Mean</th>
<th>Std. err.</th>
<th>Std. dev.</th>
<th>[95% conf. interval]</th>
</tr>
</thead>
<tbody>
<tr>
<td>mpg</td>
<td>74</td>
<td>21.2973</td>
<td>.6725511</td>
<td>5.785503</td>
<td>19.9569 22.63769</td>
</tr>
</tbody>
</table>

mean = mean(mpg)                                      t = 1.9289
H0: mean = 20                                       Degrees of freedom = 73
Ha: mean < 20                                    Ha: mean != 20
Pr(T < t) = 0.9712                                     Pr(|T| > |t|) = 0.0576
Ha: mean > 20                                    Pr(T > t) = 0.0288

The test indicates that the underlying mean is not 20 with a significance level of 5.8%.

Two-sample t test

Example 2: Two-sample t test using groups

We are testing the effectiveness of a new fuel additive. We run an experiment in which 12 cars
are given the fuel treatment and 12 cars are not. The results of the experiment are as follows:

<table>
<thead>
<tr>
<th>treated</th>
<th>mpg</th>
</tr>
</thead>
<tbody>
<tr>
<td>0</td>
<td>20</td>
</tr>
<tr>
<td>0</td>
<td>23</td>
</tr>
<tr>
<td>0</td>
<td>21</td>
</tr>
<tr>
<td>0</td>
<td>25</td>
</tr>
<tr>
<td>0</td>
<td>18</td>
</tr>
<tr>
<td>0</td>
<td>17</td>
</tr>
<tr>
<td>0</td>
<td>18</td>
</tr>
<tr>
<td>0</td>
<td>24</td>
</tr>
<tr>
<td>0</td>
<td>20</td>
</tr>
<tr>
<td>0</td>
<td>24</td>
</tr>
<tr>
<td>0</td>
<td>23</td>
</tr>
<tr>
<td>0</td>
<td>19</td>
</tr>
<tr>
<td>1</td>
<td>24</td>
</tr>
<tr>
<td>1</td>
<td>25</td>
</tr>
<tr>
<td>1</td>
<td>21</td>
</tr>
<tr>
<td>1</td>
<td>22</td>
</tr>
<tr>
<td>1</td>
<td>23</td>
</tr>
<tr>
<td>1</td>
<td>18</td>
</tr>
<tr>
<td>1</td>
<td>17</td>
</tr>
<tr>
<td>1</td>
<td>28</td>
</tr>
<tr>
<td>1</td>
<td>24</td>
</tr>
<tr>
<td>1</td>
<td>27</td>
</tr>
<tr>
<td>1</td>
<td>21</td>
</tr>
<tr>
<td>1</td>
<td>23</td>
</tr>
</tbody>
</table>

The treated variable is coded as 1 if the car received the fuel treatment and 0 otherwise.
We can test the equality of means of the treated and untreated group by typing

```
. use https://www.stata-press.com/data/r18/fuel3
. ttest mpg, by(treated)
```

Two-sample t test with equal variances

<table>
<thead>
<tr>
<th>Group</th>
<th>Obs</th>
<th>Mean</th>
<th>Std. err.</th>
<th>Std. dev.</th>
<th>[95% conf. interval]</th>
</tr>
</thead>
<tbody>
<tr>
<td>0</td>
<td>12</td>
<td>21</td>
<td>.7881701</td>
<td>2.730301</td>
<td>19.26525 22.73475</td>
</tr>
<tr>
<td>1</td>
<td>12</td>
<td>22.75</td>
<td>.9384465</td>
<td>3.250874</td>
<td>20.68449 24.81551</td>
</tr>
</tbody>
</table>

Combined

| diff  | 24  | 21.875 | .6264476 | 3.068954 | 20.57909 23.17091 |

<table>
<thead>
<tr>
<th>diff = mean(0) - mean(1)</th>
<th>( t = -1.4280 )</th>
</tr>
</thead>
<tbody>
<tr>
<td>H0: diff = 0</td>
<td>Degrees of freedom = 22</td>
</tr>
<tr>
<td>Ha: diff &lt; 0</td>
<td>Ha: diff != 0</td>
</tr>
<tr>
<td>Pr(T &lt; t) = 0.0837</td>
<td>Pr(</td>
</tr>
<tr>
<td>Ha: diff &gt; 0</td>
<td>Pr(T &gt; t) = 0.9163</td>
</tr>
</tbody>
</table>

We do not find a statistically significant difference in the means.

If we were not willing to assume that the variances were equal and wanted to use Welch’s formula, we could type

```
. ttest mpg, by(treated) welch
```

Two-sample t test with unequal variances

<table>
<thead>
<tr>
<th>Group</th>
<th>Obs</th>
<th>Mean</th>
<th>Std. err.</th>
<th>Std. dev.</th>
<th>[95% conf. interval]</th>
</tr>
</thead>
<tbody>
<tr>
<td>0</td>
<td>12</td>
<td>21</td>
<td>.7881701</td>
<td>2.730301</td>
<td>19.26525 22.73475</td>
</tr>
<tr>
<td>1</td>
<td>12</td>
<td>22.75</td>
<td>.9384465</td>
<td>3.250874</td>
<td>20.68449 24.81551</td>
</tr>
</tbody>
</table>

Combined

| diff  | 24  | 21.875 | .6264476 | 3.068954 | 20.57909 23.17091 |

<table>
<thead>
<tr>
<th>diff = mean(0) - mean(1)</th>
<th>( t = -1.4280 )</th>
</tr>
</thead>
<tbody>
<tr>
<td>H0: diff = 0</td>
<td>Welch's degrees of freedom = 23.2465</td>
</tr>
<tr>
<td>Ha: diff &lt; 0</td>
<td>Ha: diff != 0</td>
</tr>
<tr>
<td>Pr(T &lt; t) = 0.0833</td>
<td>Pr(</td>
</tr>
<tr>
<td>Ha: diff &gt; 0</td>
<td>Pr(T &gt; t) = 0.9167</td>
</tr>
</tbody>
</table>

Technical note

In two-sample randomized designs, subjects will sometimes refuse the assigned treatment but still be measured for an outcome. In this case, take care to specify the group properly. You might be tempted to let \textit{varname} contain missing where the subject refused and thus let ttest drop such observations from the analysis. Zelen (1979) argues that it would be better to specify that the subject belongs to the group in which he or she was randomized, even though such inclusion will dilute the measured effect.
Example 3: Two-sample $t$ test using variables

There is a second, inferior way to organize the data in the preceding example. We ran a test on 24 cars, 12 without the additive and 12 with. We now create two new variables, mpg1 and mpg2.

<table>
<thead>
<tr>
<th>mpg1</th>
<th>mpg2</th>
</tr>
</thead>
<tbody>
<tr>
<td>20</td>
<td>24</td>
</tr>
<tr>
<td>23</td>
<td>25</td>
</tr>
<tr>
<td>21</td>
<td>21</td>
</tr>
<tr>
<td>25</td>
<td>22</td>
</tr>
<tr>
<td>18</td>
<td>23</td>
</tr>
<tr>
<td>17</td>
<td>18</td>
</tr>
<tr>
<td>18</td>
<td>17</td>
</tr>
<tr>
<td>24</td>
<td>28</td>
</tr>
<tr>
<td>20</td>
<td>24</td>
</tr>
<tr>
<td>24</td>
<td>27</td>
</tr>
<tr>
<td>23</td>
<td>21</td>
</tr>
<tr>
<td>19</td>
<td>23</td>
</tr>
</tbody>
</table>

This method is inferior because it suggests a connection that is not there. There is no link between the car with 20 mpg and the car with 24 mpg in the first row of the data. Each column of data could be arranged in any order. Nevertheless, if our data are organized like this, \texttt{ttest} can accommodate us.

\begin{verbatim}
. use https://www.stata-press.com/data/r18/fuel
. ttest mpg1==mpg2, unpaired
Two-sample t test with equal variances

Variable | Obs | Mean | Std. err. | Std. dev. | [95% conf. interval]  
----------|-----|------|-----------|-----------|----------------------
  mpg1     | 12  | 21   | .7881701  | 2.730301  | 19.26525  22.73475  
  mpg2     | 12  | 22.75| .9384465  | 3.250874  | 20.68449  24.81551  

Combined | 24  | 21.875| .6264476  | 3.068954  | 20.57909  23.17091  

  diff    | 24  | -1.75| 1.225518  | -4.291568 | .7915684  

  diff = mean(mpg1) - mean(mpg2)  
  t = -1.4280  
  Degrees of freedom = 22

H0: diff = 0  
Ha: diff < 0  
Ha: diff != 0  
Ha: diff > 0

Pr(T < t) = 0.0837  
Pr(|T| > |t|) = 0.1673  
Pr(T > t) = 0.9163
\end{verbatim}

Paired t test

Example 4

Suppose that the preceding data were actually collected by running a test on 12 cars. Each car was run once with the fuel additive and once without. Our data are stored in the same manner as in example 3, but this time, there is most certainly a connection between the mpg values that appear in the same row. These come from the same car. The variables mpg1 and mpg2 represent mileage without and with the treatment, respectively.
ttest — t tests (mean-comparison tests)

. use https://www.stata-press.com/data/r18/fuel
. ttest mpg1==mpg2

Paired t test

<table>
<thead>
<tr>
<th>Variable</th>
<th>Obs</th>
<th>Mean</th>
<th>Std. err.</th>
<th>Std. dev.</th>
<th>[95% conf. interval]</th>
</tr>
</thead>
<tbody>
<tr>
<td>mpg1</td>
<td>12</td>
<td>21</td>
<td>.7881701</td>
<td>2.730301</td>
<td>19.26525 - 22.73475</td>
</tr>
<tr>
<td>mpg2</td>
<td>12</td>
<td>22.75</td>
<td>.9384465</td>
<td>3.250874</td>
<td>20.68449 - 24.81551</td>
</tr>
<tr>
<td>diff</td>
<td>12</td>
<td>-1.75</td>
<td>.7797144</td>
<td>2.70101</td>
<td>-3.46614 - -.038602</td>
</tr>
</tbody>
</table>

mean(diff) = mean(mpg1 - mpg2)  t = -2.2444
H0: mean(diff) = 0
Ha: mean(diff) < 0  Ha: mean(diff) != 0  Ha: mean(diff) > 0
Pr(T < t) = 0.0232  Pr(|T| > |t|) = 0.0463  Pr(T > t) = 0.9768

We find that the means are statistically different from each other at any level greater than 4.6%.

Two-sample t test compared with one-way ANOVA

Example 5

In example 2, we saw that ttest can be used to test the equality of a pair of means; see [R] oneway for an extension that allows testing the equality of more than two means.

Suppose that we have data on the 50 states. The dataset contains the median age of the population (medage) and the region of the country (region) for each state. Region 1 refers to the Northeast, region 2 to the North Central, region 3 to the South, and region 4 to the West. Using oneway, we can test the equality of all four means.

. use https://www.stata-press.com/data/r18/census
   (1980 Census data by state)
. oneway medage region

<table>
<thead>
<tr>
<th>Source</th>
<th>Analysis of variance</th>
<th></th>
<th></th>
<th></th>
<th></th>
</tr>
</thead>
<tbody>
<tr>
<td></td>
<td>SS</td>
<td>df</td>
<td>MS</td>
<td>F</td>
<td>Prob &gt; F</td>
</tr>
<tr>
<td>Between groups</td>
<td>46.3961903</td>
<td>3</td>
<td>15.4653968</td>
<td>7.56</td>
<td>0.0003</td>
</tr>
<tr>
<td>Within groups</td>
<td>94.1237947</td>
<td>46</td>
<td>2.04619645</td>
<td></td>
<td></td>
</tr>
<tr>
<td>Total</td>
<td>140.519985</td>
<td>49</td>
<td>2.8677548</td>
<td></td>
<td></td>
</tr>
</tbody>
</table>

Bartlett’s equal-variances test: chi2(3) = 10.5757  Prob>chi2 = 0.014

We find that the means are different, but we are interested only in testing whether the means for the Northeast (region==1) and West (region==4) are different. We could use oneway:

. oneway medage region if region==1 | region==4

<table>
<thead>
<tr>
<th>Source</th>
<th>Analysis of variance</th>
<th></th>
<th></th>
<th></th>
<th></th>
</tr>
</thead>
<tbody>
<tr>
<td></td>
<td>SS</td>
<td>df</td>
<td>MS</td>
<td>F</td>
<td>Prob &gt; F</td>
</tr>
<tr>
<td>Between groups</td>
<td>46.241247</td>
<td>1</td>
<td>46.241247</td>
<td>20.02</td>
<td>0.0002</td>
</tr>
<tr>
<td>Within groups</td>
<td>46.1969169</td>
<td>20</td>
<td>2.30984584</td>
<td></td>
<td></td>
</tr>
<tr>
<td>Total</td>
<td>92.4381638</td>
<td>21</td>
<td>4.40181733</td>
<td></td>
<td></td>
</tr>
</tbody>
</table>

Bartlett’s equal-variances test: chi2(1) = 2.4679  Prob>chi2 = 0.116
We could also use `ttest`:

```
. ttest medage if region==1 | region==4, by(region)
```

Two-sample t test with equal variances

<table>
<thead>
<tr>
<th>Group</th>
<th>Obs</th>
<th>Mean</th>
<th>Std. err.</th>
<th>Std. dev.</th>
<th>[95% conf. interval]</th>
</tr>
</thead>
<tbody>
<tr>
<td>NE</td>
<td>9</td>
<td>31.2333</td>
<td>.341158</td>
<td>1.023474</td>
<td>30.44662 32.02005</td>
</tr>
<tr>
<td>West</td>
<td>13</td>
<td>28.2846</td>
<td>.492357</td>
<td>1.775221</td>
<td>27.21186 29.35737</td>
</tr>
<tr>
<td>Combined</td>
<td>22</td>
<td>29.4909</td>
<td>.447306</td>
<td>2.098051</td>
<td>28.56069 30.42113</td>
</tr>
</tbody>
</table>

| diff    | 2.9487   | .659037 | 1.57399  | 4.323445 |

**diff = mean(NE) - mean(West)**

**H0: diff = 0**

**H0: diff < 0**

**H0: diff != 0**

**H0: diff > 0**

Pr(T < t) = 0.9999

Pr(|T| > |t|) = 0.0002

Pr(T > t) = 0.0001

The significance levels of both tests are the same.

### Immediate form

**Example 6**

`ttesti` is like `ttest`, except that we specify summary statistics rather than variables as arguments.

For instance, we are reading an article that reports the mean number of sunspots per month as 62.6 with a standard deviation of 15.8. There are 24 months of data. We wish to test whether the mean is 75:

```
. ttesti 24 62.6 15.8 75
```

One-sample t test

<table>
<thead>
<tr>
<th>x</th>
<th>Obs</th>
<th>Mean</th>
<th>Std. err.</th>
<th>Std. dev.</th>
<th>[95% conf. interval]</th>
</tr>
</thead>
<tbody>
<tr>
<td></td>
<td>24</td>
<td>62.6</td>
<td>3.225161</td>
<td>15.8</td>
<td>55.92825 69.27175</td>
</tr>
</tbody>
</table>

**mean = mean(x)**

**H0: mean = 75**

**H0: mean < 75**

**H0: mean != 75**

**H0: mean > 75**

Pr(T < t) = 0.0004

Pr(|T| > |t|) = 0.0008

Pr(T > t) = 0.9996
Example 7

There is no immediate form of `ttest` with paired data because the test is also a function of the covariance, a number unlikely to be reported in any published source. For unpaired data, however, we might type

```
. ttesti 20 20 5 32 15 4
Two-sample t test with equal variances
```

<table>
<thead>
<tr>
<th></th>
<th>Obs</th>
<th>Mean</th>
<th>Std. err.</th>
<th>Std. dev.</th>
<th>[95% conf. interval]</th>
</tr>
</thead>
<tbody>
<tr>
<td>x</td>
<td>20</td>
<td>20</td>
<td>1.118034</td>
<td>5</td>
<td>17.65993 22.34007</td>
</tr>
<tr>
<td>y</td>
<td>32</td>
<td>15</td>
<td>0.7071068</td>
<td>4</td>
<td>13.55785 16.44215</td>
</tr>
<tr>
<td>Combined</td>
<td>52</td>
<td>16.92308</td>
<td>0.6943785</td>
<td>5.007235</td>
<td>15.52905 18.3171</td>
</tr>
<tr>
<td>diff</td>
<td>5</td>
<td>1.256135</td>
<td>2.476979</td>
<td>7.523021</td>
<td></td>
</tr>
</tbody>
</table>

\[ \text{diff} = \text{mean}(x) - \text{mean}(y) \]
\[ t = 3.9805 \]

H0: diff = 0  
Ha: diff < 0  
Ha: diff ! = 0  
Ha: diff > 0  
Pr(T < t) = 0.9999  
Pr(|T| > |t|) = 0.0002  
Pr(T > t) = 0.0001

If we had typed `ttesti 20 20 5 32 15 4, unequal`, the test would have assumed unequal variances.

Video examples

- One-sample \( t \) test in Stata
- \( t \) test for two independent samples in Stata
- \( t \) test for two paired samples in Stata
- One-sample \( t \)-test calculator
- Two-sample \( t \)-test calculator

Stored results

`ttest` and `ttesti` store the following in `r()`:

Scalars
- `r(N_1)` sample size \( n_1 \)
- `r(N_2)` sample size \( n_2 \)
- `r(p_1)` lower one-sided \( p \)-value
- `r(p_2)` upper one-sided \( p \)-value
- `r(p)` two-sided \( p \)-value
- `r(se)` estimate of standard error
- `r(t)` \( t \) statistic
- `r(df_t)` degrees of freedom
- `r(level)` confidence level
- `r(sd_1)` standard deviation for first variable
- `r(sd_2)` standard deviation for second variable
- `r(sd)` combined standard deviation
- `r(mu_1)` mean for population 1
- `r(mu_2)` mean for population 2

Methods and formulas

See, for instance, Hoel (1984, 140–161) or Dixon and Massey (1983, 121–130) for an introduction and explanation of the calculation of these tests. Acock (2023, 165–179) and Hamilton (2013, 145–150) describe \( t \) tests using applications in Stata.
The test for $\mu = \mu_0$ for unknown $\sigma$ is given by
\[
t = \frac{\bar{x} - \mu_0}{s} \sqrt{n}
\]
The statistic is distributed as Student’s $t$ with $n-1$ degrees of freedom (Gosset [Student, pseud.] 1908).

The test for $\mu_x = \mu_y$ when $\sigma_x$ and $\sigma_y$ are unknown but $\sigma_x = \sigma_y$ is given by
\[
t = \frac{\bar{x} - \bar{y}}{\left\{ \frac{(n_x-1)s_x^2 + (n_y-1)s_y^2}{n_x + n_y - 2} \right\}^{1/2} \left( \frac{1}{n_x} + \frac{1}{n_y} \right)^{1/2}}
\]
The result is distributed as Student’s $t$ with $n_x + n_y - 2$ degrees of freedom.

You could perform `ttest` (without the `unequal` option) in a regression setting given that regression assumes a homoskedastic error model. To compare with the `ttest` command, denote the underlying observations on $x$ and $y$ by $x_j$, $j = 1, \ldots, n_x$, and $y_j$, $j = 1, \ldots, n_y$. In a regression framework, typing `ttest` without the `unequal` option is equivalent to
1. creating a new variable $z_j$ that represents the stacked observations on $x$ and $y$ (so that $z_j = x_j$ for $j = 1, \ldots, n_x$ and $z_{n_x+j} = y_j$ for $j = 1, \ldots, n_y$)
2. and then estimating the equation $z_j = \beta_0 + \beta_1 d_j + \epsilon_j$, where $d_j = 0$ for $j = 1, \ldots, n_x$ and $d_j = 1$ for $j = n_x + 1, \ldots, n_x + n_y$ (that is, $d_j = 0$ when the $z$ observations represent $x$, and $d_j = 1$ when the $z$ observations represent $y$).

The estimated value of $\beta_1$, $b_1$, will equal $\bar{y} - \bar{x}$, and the reported $t$ statistic will be the same $t$ statistic as given by the formula above.

The test for $\mu_x = \mu_y$ when $\sigma_x$ and $\sigma_y$ are unknown and $\sigma_x \neq \sigma_y$ is given by
\[
t = \frac{\bar{x} - \bar{y}}{\left( \frac{s_x^2}{n_x} + \frac{s_y^2}{n_y} \right)^{1/2}}
\]
The result is distributed as Student’s $t$ with $\nu$ degrees of freedom, where $\nu$ is given by (with Satterthwaite’s [1946] formula)
\[
\frac{\left( \frac{s_x^2}{n_x} + \frac{s_y^2}{n_y} \right)^2}{\left( \frac{s_x^2}{n_x} \right)^2 + \left( \frac{s_y^2}{n_y} \right)^2} \left( \frac{1}{n_x - 1} + \frac{1}{n_y - 1} \right)
\]
With Welch’s formula (1947), the number of degrees of freedom is given by
\[
-2 + \frac{\left( \frac{s_x^2}{n_x} + \frac{s_y^2}{n_y} \right)^2}{\left( \frac{s_x^2}{n_x} \right)^2 + \left( \frac{s_y^2}{n_y} \right)^2} \left( \frac{1}{n_x + 1} + \frac{1}{n_y + 1} \right)
\]
The test for $\mu_x = \mu_y$ for matched observations (also known as paired observations, correlated pairs, or permanent components) is given by

$$
t = \frac{\bar{d}}{s_d \sqrt{n}}
$$

where $\bar{d}$ represents the mean of $x_i - y_i$ and $s_d$ represents the standard deviation. The test statistic $t$ is distributed as Student’s $t$ with $n - 1$ degrees of freedom.

You can also use `ttest` without the `unpaired` option in a regression setting because a paired comparison includes the assumption of constant variance. The `ttest` with an unequal variance assumption does not lend itself to an easy representation in regression settings and is not discussed here.

$(x_j - y_j) = \beta_0 + \epsilon_j$.

William Sealy Gosset (1876–1937) was born in Canterbury, England. He studied chemistry and mathematics at Oxford and worked as a chemist with the brewers Guinness in Dublin. Gosset became interested in statistical problems, which he discussed with Karl Pearson and later with Fisher and Neyman. He published several important papers under the pseudonym “Student”, and he lent that name to the $t$ test he invented.

Stella Cunliffe (1917–2012) was an advocate for increased understanding of the role of human nature in experiments and methodological rigor in social statistics. She was born in Battersea, England. She was the first person from her local public girls’ school to attend college, obtaining a bachelor of science from the London School of Economics. Her first job was with the Danish Bacon Company during World War II, where she was in charge of bacon rations for London. After the war, she moved to Germany and again helped to ration food, this time for refugees. She then spent a long career in quality control at the Guinness Brewing Company. Cunliffe observed that the weights of rejected casks skewed lighter. Noting that workers had to roll casks that were too light or too heavy uphill to be remade, she had the scales moved to the top of the hill. With workers able to roll rejected casks downhill, the weight of these casks began to follow a normal distribution.

After 25 years at Guinness, Cunliffe joined the British Home Office, where she would go on to become the first woman to serve as director of statistics. During her tenure at the Home Office, she emphasized applying principles of experimental design she had learned at Guinness to the study of such topics as birthrates, recidivism, and criminology. In 1975, she became the first woman to serve as president of the Royal Statistical Society.

References


---

Also see

[R] *bitest* — Binomial probability test
[R] *ci* — Confidence intervals for means, proportions, and variances
[R] *esize* — Effect size based on mean comparison
[R] *mean* — Estimate means
[R] *oneway* — One-way analysis of variance
[R] *prtest* — Tests of proportions
[R] *sdtest* — Variance-comparison tests
[R] *ztest* — *z* tests (mean-comparison tests, known variance)
[MV] *hotelling* — Hotelling’s $T^2$ generalized means test

---

Stata, Stata Press, and Mata are registered trademarks of StataCorp LLC. Stata and Stata Press are registered trademarks with the World Intellectual Property Organization of the United Nations. Other brand and product names are registered trademarks or trademarks of their respective companies. Copyright © 1985–2023 StataCorp LLC, College Station, TX, USA. All rights reserved.